
Why Don't We Test?

Hard To Get Started

Code Too Monolithic

More Code Is More Code

Test Code Hard To Maintain

Uh, We Used To?

Tests Broke, Nobody Fixed, Turned Off

My Code's Perfect



What if ...

✦Test code was readable?
✦Tests were concise?

✦Test reports were useful?
✦Failures were well described?
✦Mocking was easy?

… wouldn't that be most logical?



Spock
✦Groovy DSL for Testing
✦Builds on JUnit 4
✦Goals: Concise, Maintainable
✦Given / When / Then
✦Authors:
✦Peter Niederwieser
✦Luke Daley



My Path

1990's Wrote own test framework –  in PL/1 
5000 - 7000 tests

2001 First JUnit tests (Tapestry template parser)

2003 Started using TestNG, EasyMock

2005 Started using Selenium 1

2006 Dabbled in Groovy

2010 Spock! ~ 0.4

2012 Rewrote ~650 tests TestNG ➠ Spock (tapestry-ioc)



Feature Testing



Terminology

Specification

Feature

Fixture

Feature
Collaborator

Collaborator

System 
Under 

Specification



Sky.java

First Specification
SkySpecification.groovy
package org.example

import spock.lang.*

class SkySpecification extends Specification {

  ...
} All Specifications extend from this base class

package org.example.sus;

public class Sky {
  public String getColor() {
    return "blue";
  }
}

spock 0.7-
groovy-2.0



Feature Methods

def "sky is blue"() {
  setup:
  def sky = new Sky()
  
  expect:
  sky.color == "blue"
}



Execution
$ gradle clean test
:clean
:compileJava
:compileGroovy
:processResources UP-TO-DATE
:classes
:compileTestJava UP-TO-DATE
:compileTestGroovy
:processTestResources
:testClasses
:test

BUILD SUCCESSFUL

Total time: 12.106 secs
~/workspaces/github/spock-examples
$

$ tree build/reports/
build/reports/
└── tests
    ├── base-style.css
    ├── css3-pie-1.0beta3.htc
    ├── index.html
    ├── org.example.SkySpecification.html
    ├── org.example.html
    ├── report.js
    └── style.css

1 directory, 7 files
~/workspaces/github/spock-examples
$









sky.color == "green"







Feature Method Blocks
def "sky is blue"() {
  setup:
  def sky = new Sky()
  
  expect:
  sky.color == "blue"
}

✦ setup: or given: 

✦ expect:

✦ when:

✦ then:

✦ where:

✦ cleanup:

✦ Feature methods must contain at least one block
otherwise, not a feature method



setup:

✦ Initialization of the fixture
✦Always at top of method
✦Can't be repeated
✦Can be given: instead

✦Anything up to first label is implicit setup:



expect:

✦Combines stimulus and response
✦Contains only conditions and variable definitions
✦Conditions assert Groovy truth

✦Use with pure functions
✦No side effects, just inputs and result

expect:
  sky.color == "blue"

Stimulus

Response



when: / then:
  def "clouds are grey"() {
    def sky = new Sky()

    when:
    sky.addStormSystem()

    then:
    sky.color == "grey"
  }

✦Tests code with side effects
✦then: may only contain:

✦conditions
✦exception conditions
✦ interactions
✦variable definitions

Stimulus

Response



when: / then:
  def "clouds are grey"() {
    def sky = new Sky()

    when:
    sky.addStormSystem()

    then:
    sky.color == "grey"
  }



Test driven
Sky.java
package org.example.sus;

public class Sky {

  private String color = "blue";

  public String getColor() {
    return color;
  }

  public void addStormSystem() {
    color = "grey";
  }
}



then: Old Values

def "pushing an element on the stack increases its size by one"() {
  def stack = new Stack()

  when:

  stack.push("element")

  then: 

  stack.size() == old(stack.size()) + 1
}

Expression value captured before where: block



Extended Assert
  def "use of extended assert"() {

    expect:
    assert 4 == 5, "Big Brother says there are four fingers"
  }



cleanup:
setup:
def file = new File("/some/path")
file.createNewFile()

// ...

cleanup:
file.delete()

✦Cleanup external resources
✦Always invoked, even if previous exceptions



where:

✦Parameterizes feature method with data
✦Must be last block
✦Can use | or || as separator

def "length of crew member names"() {
  expect:
  name.length() == length

  where:
  name     | length
  "Spock"  | 5
  "Kirk"   | 4
  "Scotty" | 6
}



where:

One entry for three 
feature method 

executions



where: using lists
  def "length of crew member names (using lists)"() {
    expect:
    name.length() == length

    where:
    name << ["Spock", "Kirk", "Scotty"]
    length << [5, 4, 6]
  }

Could come from external file or database

where:
[name, age, gender] = sql.execute("select name, age, sex from ⏎ 
 customer")



class CronExpressionSpec extends Specification {

  def propertyMissing(String name) { Calendar[name] }

  def "test CRON expressions for validity"() {

    def cal = Calendar.getInstance();

    def exp = new CronExpression(expr)

    cal.set year, month, day, hour, minute, second

    expect:

    exp.isSatisfiedBy(cal.time) == satisfied

    where:

    expr                    | year | month   | day | hour | minute | second | satisfied
    "0 15 10 * * ? 2005"    | 2005 | JUNE    | 1   | 10   | 15     | 0      | true
    "0 15 10 * * ? 2005"    | 2006 | JUNE    | 1   | 10   | 15     | 0      | false
    "0 15 10 * * ? 2005"    | 2005 | JUNE    | 1   | 10   | 16     | 0      | false
    "0 15 10 * * ? 2005"    | 2005 | JUNE    | 1   | 10   | 14     | 0      | false
    "0 15 10 L-2 * ? 2010"  | 2010 | OCTOBER | 29  | 10   | 15     | 0      | true
    "0 15 10 L-2 * ? 2010"  | 2010 | OCTOBER | 28  | 10   | 15     | 0      | false
    "0 15 10 L-5W * ? 2010" | 2010 | OCTOBER | 26  | 10   | 15     | 0      | true
    "0 15 10 L-1 * ? 2010"  | 2010 | OCTOBER | 30  | 10   | 15     | 0      | true
    "0 15 10 L-1W * ? 2010" | 2010 | OCTOBER | 29  | 10   | 15     | 0      | true
  }
}

Pretty formatting care of IntelliJ



where: derived values
  def "length of crew member names (with derived values)"() {
    expect:
    name.length() == length

    where:
    name << ["Spock", "Kirk", "Scotty"]
    length = name.length()
  }



class TimeServiceSpec extends Specification {

    def ts = new TimeServiceImpl()

    static final int second = 1000,
                     minute = 60 * second,
                     hour = 60 * minute,
                     day = 24 * hour,
                     bigTime = (3 * day) + (2 * hour) + (10 * minute) + (5 * second)

    @Unroll
    def "formatInterval #delta ms, #approximate should be '#expected'"() {

        expect:

        ts.formatInterval(start, end, approximate) == expected

        where:

        delta   | approximate || expected

        0       | false       || "now"
        999     | false       || "now"
        -999    | false       || "now"
        1000    | false       || "in one second"
        -3000   | false       || "three seconds ago"
        bigTime | true        || "in about three days and two hours"

        now = System.currentTimeMillis()
        start = new Date(now)
        end = new Date(now + delta)
    }
}

Computed for 
each iteration



Block labels
  def "clouds are grey"() {

    given: "A fresh Sky"
    def sky = new Sky()

    when: "A storm system rolls in"
    sky.addStormSystem()

    then: "It all goes grey"
    sky.color == "grey"
  }

✦Allowed, but not (currently) used
✦and: "block" allowed, does nothing



Beyond Feature Methods



Fields
package org.example

import org.example.sus.Sky
import spock.lang.Specification

class SkySpecification extends Specification {

  def sky = new Sky()

  def "sky is blue by default"() {
    expect:
    sky.color == "blue"
  }

  def "clouds are grey"() {

    given: "A fresh Sky"

    when: "A storm system rolls in"
    sky.addStormSystem()

    then: "It all goes grey"
    sky.color == "grey"
  }
}

New for each feature method



Shared Fields

class MySpecification extends Specification {
  @Shared def resource = new AnExpensiveResource()

  static final PASSWORD = "sayfriendandenter"
  … 
}

Created once, shared across all instances

Statics should be final and immutable



Fixture Methods
def setup() { … }

def cleanup() { … }

✦Create / initialize instance of Specification
✦ Invoke setup()
✦ Invoke feature method
✦ Invoke cleanup()



Fixture Methods

✦ Instance created for Specification setup / cleanup
✦May only access @Shared and static fields

def setupSpec() { … }

def cleanupSpec() { … }



Fixture Method Inheritance
abstract class AbstractSharedRegistrySpecification extends Specification {

  static Registry registry

  def methodMissing(String name, args) {
    registry."$name"(* args)
  }

  def setupSpec() {
    if (registry == null) {
      registry = IOCUtilities.buildDefaultRegistry()
    }
  }

  def cleanupSpec() {
    registry.cleanupThread();
  }

}

class AspectInterceptorBuilderImplSpec extends AbstractSharedRegistrySpecification {

  @Shared
  private AspectDecorator decorator

  def setupSpec() {
    decorator = getService AspectDecorator
  }

Base class setupSpec() invoked first

Sublass setupSpec() invoked after base class

methodMissing() ➠ registry.getService()



Exception Conditions
def ins = new ClassInstantiatorImpl(ContextCatcher, ⏎
  ContextCatcher.constructors[0], null)

def "may not add a duplicate instance context value"() {

  given:
  def ins2 = ins.with(String, "initial value")

  when:
  ins2.with(String, "conflicting value")

  then:
  IllegalStateException e = thrown()

  e.message == "An instance context value of type java.lang.String ⏎
  has already been added."
}

Or: def e = thrown(IllegalStateException)



notThrown()
def "HashMap accepts null key"() {
  setup:
  def map = new HashMap()
  
  when:
  map.put(null, "elem")
  
  then:
  notThrown(NullPointerException)
}

Documentation value only

Also: noExceptionThrown()



Mocks and Interactions



Specification

Feature

Fixture

Feature
Collaborator

Collaborator

SystemUnder 
Specification



Configured 
Instance

Mock Object

System 
Under 

Specification



CustomerDAOPayment 
Processor

PaymentProcessor.groovy
package org.example.sus

class PaymentProcessor {

  CustomerDAO customerDAO

  def applyPayment(long customerId, BigDecimal amount) {

    Customer customer = customerDAO.getById(customerId)

    if (customer == null)
      throw new IllegalArgumentException("No customer #$customerId")

    customer.accountBalance += amount

    customerDAO.update(customer)
  }
}

CustomerDAO.java
package org.example.sus;

public interface CustomerDAO {

  Customer getById(long id);

  void update(Customer customer);
}



class ApplyPaymentSpecification extends Specification {

  CustomerDAO dao = Mock()
  PaymentProcessor processor

  def setup() {
    processor = new PaymentProcessor(customerDAO: dao)
  }

  … 
}|

Factory method



  def "unknown customer id is an exception"() {
    when:
    processor.applyPayment(12345, 100)

    then:
    1 * dao.getById(12345) >> null

    IllegalArgumentException e = thrown()

    e.message == "No customer #12345"
  }

Define behavior for 
preceding when: block

Defining Mock Behavior



  def "valid customer id for update"() {
    when:
    processor.applyPayment(customer.id, 200)

    then:
    1 * dao.getById(customer.id) >> customer
    1 * dao.update(customer)

    customer.accountBalance == 500

    where:
    customer = new Customer(id: 98765, accountBalance: 300)
  }



1 * dao.getById(12345) >> null
Number of invocations: 

cardinality

Argument Constraints

Returns a value

Target  and Method Constraints



Cardinality
1 * dao.getById(12345) >> null

Cardinality Description
(_..n) * target.method(…) Up to n times
n * target.method(…) Exactly n times
(n.._) * target.method(…) At least n times

Omitted Optional interaction, must have 
return value 



Argument Constraints
1 * dao.getById(12345) >> null

Argument Constraint Description
_     (underscore) Any argument
*_ Any number of arguments
!null Any non-null argument
value Argument equals value
!value Argument does not equal value
_ as Type Non-null, assignable to type
{ it -> … } Closure: validates, returns true / false



class InteractionsSpecification extends Specification {

  Operation mock = Mock()

  interface Operation {
    Object op(input)
  }

  def isOdd(input) { input % 2 != 0 }

  def "closure for parameter constraint"() {
    when:
    assert mock.op(3) == 6
    assert mock.op(7) == 14

    then:
    (2..7) * mock.op({ isOdd(it) }) >> { 2 * it }
  }

Closures for Argument Contraints

Return value computed by closure

Helper method

it is list of arguments



  def "wrong number of invocations"() {
    when:
    assert mock.op(7) == 14

    then:
    (2..7) * mock.op({ isOdd(it) }) >> { 2 * it }
  }

Too few invocations for:

(2..7) * mock.op({ isOdd(it) }) >> { 2 * it }   (1 invocation)

Unmatched invocations (ordered by similarity):

None

& at org.spockframework.mock.runtime.InteractionScope.verify… 



  def "parameter does not match closure constraint"() {
    when:
    assert mock.op(3) == 6
    assert mock.op(4) == null
    assert mock.op(7) == 14

    then:
    _ * mock.op({ isOdd(it) }) >> { 2 * it }
  }

Mocks are Lenient



Less Lenient Mocks
  def "detecting parameter that doesn't match"() {
    when:
    assert mock.op(3) == 6
    assert mock.op(4) == null
    assert mock.op(7) == 14

    then:
    _ * mock.op({ isOdd(it) }) >> { 2 * it }

    0 * _
  }

Too many invocations for:

0 * _   (1 invocation)

Matching invocations (ordered by last occurrence):

1 * mock.op(4)   <-- this triggered the error

& at org.spockframework.mock.runtime.MockInteraction… 

aka "any interaction"



Simplicity

Consistency

Efficiency

Feedback



Target Constraints
1 * dao.getById(12345) >> null

Target Constraint Description

name Match the named mock

_  (underscore) Match any mock



Method Constraints
1 * dao.getById(12345) >> null

Method Constraint Description

name Match method with name

_  (underscore) Match any method

/regexp/ Match any method that matches the expression

e.g. bean./set.*/(_)



Return Values
1 * dao.getById(12345) >> null

Return Value Constraint Description

>> value Return the value

>> { … } Evaluate the closure, passed list of arguments, 
return result

>>> [ value1, value2, …] Return the values; final value is repeated

May be an Iterable type

Need new slide about closure

closure w/ 1 arg recieves list of args

closure w/ N args recieves actual arguments



Chained Return Values

then:
  service.getStatus() 
    >>> ["ok", "ok", "fail"]

  >>  { throw new RuntimeException("Status failure."); }

The last value sticks



Ordered Interactions

def "test three amigos"() {
  when:
  facade.doSomething()

  then:
  1 * collab1.firstPart()
  1 * collab2.secondPart()

  then:
  1 * collab3.thirdPart()
}

No order checking 
on firstPart(), 
secondPart()

thirdPart() only 
allowed after both 

firstPart(), 
secondPart()



Stubbing

def "frobs the gnop"() {

  checker.isValid(_) >> true

  when:
  … 
}

A "global interaction" 
valid to end of 

method



Extensions



@Unroll
  @Unroll
  def "Crew member '#name' length is #length"() {
    expect:
    name.length() == length

    where:
    name     | length
    "Spock"  | 5
    "Kirk"   | 4
    "Scotty" | 6
  }





@Unroll
class CanonicalWhereBlockExampleSpecification extends Specification {

  def "Crew member '#name' length is #length"() {
    … 
  }

  def "length of crew member names (using lists)"() {
    … 
  }

  def "length of crew member names (with derived values)"() {
    … 
  }
}



@Timeout
@Timeout(5)
def "can access data in under five seconds"() {
  … 
}

@Timeout(value=100, unit=TimeUnit.MILLISECONDS)
def "can update data in under 100 ms"() {
  … 
}

✦Feature and fixture methods run on main thread
✦A second thread may interrupt the main thread

✦Can be placed on class to affect all feature methods



@Stepwise

✦Feature methods run in declaration order
✦Failed methods cause remainder to be skipped

@Stepwise
class BuildingBlocksSpecification extends Specification {

  def "first step()" { … }

  def "second step"() { … }

  def "third step"() { … }



@AutoCleanup

✦Will invoke close() on field's value

✦Exceptions are reported but are not failures
✦quiet=true ➠ Don't report exceptions
✦value attribute is name of method to invoke

def DatabaseSpecification … {

  @AutoCleanup @Shared
  Connection connection

  … 
} 



@Ignore / @IgnoreRest
✦Temporarily control which feature methods execute
✦@Ignore ➠ Ignore this method, run others
✦@IgnoreRest ➠ Run this method, ignore others
✦ Ignored Methods Reported



More 
Information



Other Spock Topics

✦Stubs
✦Spies
✦Groovy Mocks
✦Creating Extensions
✦Asynchronous Testing
✦Hamcrest Assertions



https://github.com/spockframework/spock



http://docs.spockframework.org



https://github.com/hlship/spock-examples



http://howardlewisship.com



http://www.flickr.com/photos/28687962@N08/4094051518

http://xkcd.com/703/

http://www.flickr.com/photos/41254175@N07/5878491555

http://www.flickr.com/photos/nolifebeforecoffee/124659356/

http://www.flickr.com/photos/oldpatterns/4561597023/

http://www.flickr.com/photos/marcobellucci/3534516458/

http://www.flickr.com/photos/thomashawk/4958821586

